

- o Este teste é sem consulta e de realização individual.
- o É proibida a utilização de computadores, máquinas de calcular e telemóveis.
- o Resolva cada grupo numa folha separada (note-se que folhas e páginas não são a mesma coisa!) utilizando a linguagem Visual Basic
- o Coloque o Nome e Número em todas as folhas
- o Duração: Alunos Ordinários – 105 min, Alunos com dispensa de Avaliação Contínua – 135 min

### Grupo 1

(só para alunos com dispensa de avaliação contínua ou melhoria)

Elabore um programa que lei uma sequência de números inteiros e apresente numa ListBox todos os números pares lidos, pela ordem inversa da sua inserção. O número de elementos a ler é dado pelo utilizador e limitado a 100 unidades. Recomenda-se a utilização de um vector para resolver este problema.

### Grupo 2

Considere uma matriz de números reais que representa as precipitações num conjunto de estações meteorológicas ao longo de um mês, em que as linhas correspondem às estações e as colunas aos dias. O número máximo de estações é 24 e o número de dias do mês varia entre 28 e 31 dias.

- Elabore uma subrotina que leia para uma matriz os valores da precipitação numa estação meteorológica (número da estação passado como parâmetro) ao longo do mês. Note que os valores da precipitação são sempre positivos.
- Elabore uma função que retorne o total de precipitação ao longo do mês numa dada estação (número da estação é passado como parâmetro).
- Elabore uma subrotina que, utilizando a função da alínea b), retorne a estação com mais precipitação e o respectivo valor.
- Elabore um programa que peça ao utilizador o número de estações e o número do mês a tratar e, utilizando as rotinas acima, leia as precipitações ao longo do mês nas estações. O programa deve apresentar numa ListBox as precipitações lidas e em duas TextBox a estação com maior precipitação e o respectivo valor.

### Grupo 3

a) Elabore uma função que recebe uma frase (*string*) e um número, e retorne uma frase (*string*) que tenha por base a frase original, em que sejam eliminadas todas as palavras da frase com um número de vogais que seja divisor do número dado, com a excepção da unidade. Não deve ser feita qualquer distinção entre maiúsculas e minúsculas e os espaços repetidos entre as palavras podem também ser eliminados.


Nota: Não será cotada qualquer solução que não implemente uma função.

Exemplo: “Ab desef2i cBa dekefia cbAokl i abeidoUa” e 6 --> “Ab cBa dekefia i”

b) Elabore um programa que permita ao utilizador indicar um número e que leia uma sequência de frases (terminada por uma *string* vazia). Para cada uma das frases lidas, o programa deve apresentar a frase original numa *listbox* e a frase transformada pela função da alínea anterior noutra.

	Alunos Ordinários	Alunos c/ Dispensa
Grupo	Cotação (valores)	Cotação (valores)
I	-	5
II	13	10
III	7	5
Total	20	20

## Grupo 2



Option Explicit

```
Const MAX_EST = 24
```

```
Private Sub cmdLer_Click()  
    Dim mp(MAX_EST, 31) As Single, nEst As Integer, mes As Integer  
    Dim nDias As Integer, x As Integer, y As Integer  
    Dim linha As String, emaxp As Integer, maxp As Single  
  
    nEst = Val(txtNE.Text)  
    mes = Val(txtMes.Text)  
  
    If nEst < 1 Or nEst > MAX_EST Or mes < 1 Or mes > 12 Then  
        MsgBox "Valores errado, o programa vai terminar!"  
        Exit Sub  
    End If  
  
    If mes = 2 Then  
        nDias = 28  
    ElseIf mes = 4 Or mes = 6 Or mes = 9 Or mes = 11 Then  
        nDias = 30  
    Else  
        nDias = 31  
    End If  
  
    For x = 1 To nEst  
        Ler mp, x, nDias  
    Next  
  
    For x = 1 To nEst  
        linha = x & " : "  
        For y = 1 To nDias  
            linha = linha & " " & mp(x, y)  
        Next  
        lstP.AddItem linha  
    Next  
    MaisP mp, nEst, nDias, emaxp, maxp  
    txtEMaxP.Text = emaxp  
    txtMaxP.Text = maxp  
End Sub  
  
Sub Ler(m() As Single, ByVal l As Integer, ByVal nc As Integer)  
    Dim x As Integer  
  
    For x = 1 To nc  
        Do  
            m(l, x) = Val(InputBox("Precipitação no dia " & x))  
        Loop While m(l, x) < 0  
    Next
```

```

End Sub

Function TotP(m() As Single, ByVal l As Integer, ByVal nc As Integer)
    Dim x As Integer, s As Single


    For x = 1 To nc
        s = s + m(l, x)
    Next
    TotP = s
End Function

Sub MaisP(m() As Single, ByVal nl As Integer, ByVal nc As Integer, _
    maxE As Integer, maxp As Single)
    Dim x As Integer, aux As Single

    maxp = TotP(m, 1, nc)
    maxE = 1
    For x = 2 To nl
        aux = TotP(m, x, nc)
        If aux > maxp Then
            maxp = aux
            maxE = x
        End If
    Next
End Sub

```

### Grupo 3



Option Explicit

```

Private Sub cmdConverter_Click()
    Dim Frase As String, FraseTransform As String, Numero As Integer
    Numero = Val(txtNumero.Text)
    Frase = InputBox("Introduza a Frase (Enter para terminar)")
    Do While Frase <> ""
        lstFraseOriginal.AddItem Frase
        FraseTransform = ConverterFrase(Frase, Numero)
        lstFraseTransform.AddItem FraseTransform
        Frase = InputBox("Introduza a Frase (Enter para terminar)")
    Loop
End Sub

```

```

Private Function ConverterFrase(ByVal Frase As String, ByVal Numero As Integer) As String
    Dim PosEspaco As Integer, Palavra As String, i As Integer, ContaVogais As Integer, _
        Caracter As String, NovaFrase As String
    Frase = Trim(Frase)
    Do While Frase <> ""
        PosEspaco = InStr(Frase, " ")
        If PosEspaco <> 0 Then
            Palavra = Left(Frase, PosEspaco - 1)
            Frase = Trim(Right(Frase, Len(Frase) - PosEspaco))
        Else
            Palavra = Frase
            Frase = ""
        End If
        ContaVogais = 0
        For i = 1 To Len(Palavra)
            Caracter = LCase(Mid(Palavra, i, 1))
            If Caracter = "a" Or Caracter = "e" Or Caracter = "i" _
                Or Caracter = "o" Or Caracter = "u" Then
                ContaVogais = ContaVogais + 1
            End If
        Next
        If ContaVogais = 1 Or ContaVogais Mod Numero <> 0 Then
            NovaFrase = NovaFrase + Palavra + " "
        End If
    Loop
    ConverterFrase = RTrim(NovaFrase)
End Function

```